APL im Mathematikunterricht

Mag. Gottfried Bach, Wien

Abstract:

Die Idee des Experimentierens, wie es in den naturwissenschaftlichen Fächern üblich ist, hat in den Mathematikunterricht kaum Eingang gefunden.

Im Vortrag soll - hauptsächlich an Beispielen - gezeigt werden, wie APL (eine mathematische Notation und Programmiersprache) im Mathematikunterricht eingesetzt wurde: als Werkzeug, als Hilfe beim Erlernen mathematischer Zusammenhänge, der Computer als "mathematisches Labor".

Vortrag:

APL (A Programming Language) gilt heute als die mächtigste Programmiersprache und ist sowohl im kaufmännischen als auch im technisch-wissenschaftlichen Bereich weit verbreitet. Wie andere Sprachen auch, wird APL im Lehrwesen an Schulen und Universitäten vielfach eingesetzt

- im Verwaltungsbereich,
- im Informatikstudium,
- zu Prüfungsdurchführungen und Ergebnisauswertungen,
- im computerunterstützten Unterricht u.a.

Von all dem soll in diesem Vortrag nicht die Rede sein. Hier soll gezeigt werden, wie APL im Mathematikunterricht verwendet werden kann bzw. verwendet wurde zur Durchführung von "Experimenten".

A. Etwas APL

Die Eingabe ist immer 6 Stellen eingerückt, die Antwort vom Computer erfolgt in der nächsten Zeile am linken Rand.

```
      3+2                          Addieren von Zahlen (Skalaren)
5
      2-5                          Subtrahieren
¯3

      3+7 5 9
10 8 12

      1 2 3+4 5 6                  Addieren von Listen (Vektoren)
5 7 9

      X←3                          X "ist" 3 oder

      Y←4                          Y "wird zu" 4

      X×Y                          X mal Y
12

      3=4                          3 ist gleich 4
0                                  nein oder falsch
```

```
      7<8                                   7 ist kleiner als 8
1                                           ja oder richtig

      3<1 5 2 8 9
0 1 0 1 1                                   Boole'scher (oder logischer) Vektor

      0 1 0 1 1/1 5 2 8 9                   Kompression. Auswählen der Elemente im
5 8 9                                       rechten Argument, die einer 1 im linken
                                            Argument entsprechen.

      V←1 5 2 8 9

      3<V
0 1 0 1 1

      (3<V)/V                               Aus V die Elemente auswählen, die
5 8 9                                       größer als 3 sind.

      ρV                                    Anzahl der Elemente in V
5
```

Die Funktionen können mit Hilfe einer "Funktionsmaschine" veranschaulicht werden.



Eine Funktionsmaschine mit 2 Eingängen (Argumenten): eine zweiwertige Funktion.

α ist das linke Argument
ω ist das rechte Argument

Zum Beispiel die "+"-Maschine



"+" ist eine zweiwertige Funktion mit einem expliziten Resultat.

Die ρ -Maschine:



"ρ" ist eine einwertige Funktion mit einem expliziten Resultat.

Weitere Funktionsbeispiele:

```
      1∧1
1
```
logisches "und"

```
      0∧1
0
```
" "

```
      3⌊5
3
```
min (3,5)

```
      8 9⌈4 13
8 13
```
paarweises Maximum

```
      2*3
8
```
Potenzieren

```
      2⍟8
3
```
Logarithmus Basis 2

```
      2*2⍟8
8
```
$f^{-1}(f(x))$




```
      2⍟2*3
3
```

```
      ⍳5
1 2 3 4 5
```
Indexvektor bilden

```
      +/⍳5
15
```
Σ

```
      ?6
1
```
Zufallszahl aus 1 bis 6

```
      ?6 6 6 6
5 3 4 2
```
4 Zufallszahlen aus 1 bis 6

```
      10⍴6
6 6 6 6 6 6 6 6 6 6
```
10 Sechser

```
      ?10⍴6
1 5 5 6 3 4 5 1 1 4
```
10 Zufallszahlen von 1 bis 6

B. Einige "Experimente"

1. Wahrscheinlichkeitsberechnung

WERFEN EINER MUENZE

`?2` `2`	Zufallszahl 1 oder 2
`?2 2 2 2 2` `1 1 1 1 2`	5 Zufallszahlen "KOPF" oder "ADLER"
`?10ρ2` `2 2 2 2 1 2 1 2 2 2`	10 "KOPF" oder "ADLER"
`1=?10ρ2` `1 1 0 1 0 0 0 1 1 0`	
`+/1=?10ρ2` `4`	4 mal "KOPF"
$\boxed{\texttt{+/1=?Nρ2}}$	Anzahl "KOPF" bei N Würfen

Eine selbstdefinierte Funktion (ein Programm)

$\boxed{\begin{array}{l}\textit{DEFINE}\\ \textit{MUENZE:+/1=?ωρ2}\end{array}}$	ω wird immer als Name für das rechte Argument verwendet, α für das linke.



Die Funktionsmaschine "MUENZE"

`MUENZE 100` `51`	Münze 100 mal geworfen
`MUENZE 1000` `504`	1000 mal



Würfel Experiment:

Wie groß ist die Wahrscheinlichkeit, mit 2 Würfeln die Augensumme 8 zu werfen?

`      ?6` `1`	Rollen eines Würfels
`      ?10⍴6` `5 3 4 2 1 5 5 6 3 4`	10 mal
`      W1←?10⍴6`	Erster Würfel 10 mal
`      W2←?10⍴6`	Zweiter Würfel 10 mal
`      W1` `5 1 1 4 5 1 3 1 3 5`	
`      W2` `4 6 6 4 1 4 3 5 6 5`	
`      W1←?100⍴6`	Erster Würfel 100 mal gerollt
`      W1` `2 1 5 2 4 5 6 3 2 6 5 5 ...`	
`      W2←?100⍴6`	Zweiter Würfel 100 mal gerollt
`      W2` `1 5 3 3 3 1 4 6 4 6 4 1 ...`	
`      W←W1+W2`	Paarweise Addition
`      W` `3 6 8 5 7 6 10 9 6 12 9 ...`	Augensummen
`      8=W` `0 0 1 0 0 0 0 0 0 0 0 0 ...`	Augensumme ist gleich 8
`      +/8=W` `13`	13 Achter
`DEFINE` `WUERFEL: +/8=(?ω⍴6)+(?ω⍴6)`	Als selbstdefinierte Funktion (Programm)

```
      WUERFEL 100
20

      WUERFEL 1000
146

      WUERFEL 10000
1396
```



Nach der "experimentellen" Wahrscheinlichkeit zur "mathematischen" Wahrscheinlichkeit.

Hier hilft das "Produkt" aller Paare von Elementen zweier Vektoren:

```
      2 3 ∘.× 3 2 5
6  4 10
9  6 15
```

```
      3 5 4 ∘.≤ 6 3 4
```

≤	6	3	4
3	1	1	1
5	1	0	0
4	1	0	1

```
      A←ι6

      A∘.+A
2  3  4  5  6  7
3  4  5  6  7  8
4  5  6  7  8  9
5  6  7  8  9 10
6  7  8  9 10 11
7  8  9 10 11 12
```

Die Summe aller Paare

5 von 36 Möglichkeiten sind günstig:

$W(8) \approx \frac{5}{36} \approx 0,14$

Leider treten folgende Ergebnisse bei guter Software kaum auf:

```
      WUERFEL 10
0

      WUERFEL 1000
0

      WUERFEL 1000
900
```

Aber schon bei geringen Abweichungen von der "erwarteten" Wahrscheinlichkeit gibt es Fragen nach dem Bereich, in dem die Antwort liegen sollte. Hier schließen sich sehr spontan Überlegungen z.B. über das Schwankungsmaß nach Bernoulli an, wobei bei WUERFEL 1000

die Schwankung $s = \sqrt{n \cdot w \cdot (1-w)} = \sqrt{1000 \cdot \frac{5}{36} \cdot \frac{31}{36}} = 11$

wäre und damit $129 < \text{Antwort} < 151$.

ZAHLENTHEORIE

```
      3|5
2
```

5 = 3.Q + REST
Rest modulo 3

```
      3|5 7 8
2 1 2

      3|ι10
1 2 0 1 2 0 1 2 0 1
```

Die Dreier-Reste der Zahlen von 1 bis 10

```
      0=3|ι10
0 0 1 0 0 1 0 0 1 0
```

Wo sind die Dreier-Reste gleich 0?

```
      (0=3|ι10)/ι10
3 6 9
```

Von ι10 diejenigen auswählen, deren Dreier-Reste 0 sind, das heißt, die durch 3 teilbar sind.

Alle Teiler einer Zahl finden:

```
      1 2 3|3
0 1 0
```

Einser-, Zweier-, Dreier-Rest von 3.

```
      (ι10)|10
0 0 1 2 0 4 3 2 1 0
```

Einser- bis Zehner-Rest von 10.

```
      RESTE←(ι10)|10

      0=RESTE
1 1 0 0 1 0 0 0 0 1
```

Wo ist der Rest 0.

```
      (0=RESTE)/ι10
1 2 5 10
```

Diejenigen Elemente von ι10 auswählen, wo der Rest 0 ist, also die Teiler von 10.

```
      (0=(ι10)|10)/ι10
1 2 5 10
```

```
DEFINE
TEILER: (0=(ιω)|ω)/ιω
```

Die Teiler-Funktion

```
      TEILER 5
1 5

      TEILER 10
1 2 5 10

      TEILER 12
1 2 3 4 6 12


      TEILER 37
1 37
```

```
      ρ TEILER 37
2
```

Anzahl der Teiler

```
      2=ρ TEILER N
```

Anzahl der Teiler einer Zahl N ist gleich 2: ja oder nein.

```
DEFINE
PRIM: 2=ρ TEILER ω
```

Die PRIM-Funktion

```
      PRIM 37
1
```

37 ist prim.

```
      PRIM 12
0
```

12 ist nicht prim.

```
    DEFINE
    PERFEKT: (2×ω)=+/TEILER ω
```

Die Summe der Teiler einer Zahl ist gleich der doppelten Zahl.

```
      PERFEKT 6
1
```

6 ist eine perfekte Zahl.

```
      PERFEKT 50
0

      PERFEKT 28
1
```

ElementareZahlentheorie ist eine Fundgrube für interessante Experimente.

MENGENLEHRE

A



B



```
      A←3 7 1 2 9
```

Menge A

```
      B←4 1 6 9 5 8
```

Menge B

```
      3∊4 3 7
1
```

3 ist ein Element von 4 3 7

```
      3∊B
0
```

3 ist ein Element von B
nein

```
      3 4 5∊B
0 1 1
```

3 4 5 enthalten in B
nein, ja, ja

```
      A∊B
0 0 1 0 1
```

A enthalten in B
elementweise geprüft

```
      B∊A
0 1 0 1 0 0
```

```
      ∧/1 1 1 1 1
1
      ∧/1 0 1 1 1
0
      ∧/A∈B
0
```

Die "und"-Reduktion gibt genau dann 1 wenn alle Elemente 1 sind.

```
DEFINE
TEILMENGE: ∧/α∈ω
```




```
      A TEILMENGE B
0
```




```
      A
3 7 1 2 9
      B
4 1 6 9 5 8
      A∈B
0 0 1 0 1
      0 0 1 0 1/A
1 9
```

Wähle aus A diejenigen Elemente aus, für die gilt, daß A in B.

oder wähle aus A diejenigen Elemente aus, die (auch) in B sind.

```
DEFINE
DURCHSCHNITT: (α∈ω)/α
```

```
      A DURCHSCHNITT B
1 9
```

FOLGEN UND REIHEN

```
      ⍳6
1 2 3 4 5 6
      3×⍳6
3 6 9 12 15 18
      1+3×⍳6
4 7 10 13 16 19
      2*⍳10
2 4 8 16 32 64 128 256 512 1024
      ÷2*⍳10
0.5 0.25 0.125 0.0625 0.03125 0.015625 0.0078125 0.00390625 0.001953125 0.0009765625
```

Die ersten 10 Potenzen von 2.

Deren Reziprokwerte

Die Summe der Reziprokwerte

```
      +/÷2*⍳10
0.9990234375
```

```
DEFINE
REIHE1:+/÷2*⍳ω
```

```
      REIHE1 10
0.9990234375

      REIHE1 20
0.9999990463

      REIHE1 100
1
```

konvergiert sehr schnell

```
      ⍳10
1 2 3 4 5 6 7 8 9 10

      ÷⍳10
1 0.5 0.3333333333 0.25 0.2 0.1666666667 0.1428571429 0.125 0.1111111111 0.1
```

Die Harmonische Reihe

```
DEFINE
HARMONISCH: +/÷⍳ω
```

```
      HARMONISCH 10000
9.787606036

      HARMONISCH 100000
12.09014613

      HARMONISCH 1000000
14.39272672
```

Dieses Beispiel ist eher ein "Anti-Experiment": Die Summe wächst so langsam, daß fast alle Schüler jede Wette auf Konvergenz eingehen. Die Summe wird meist bei circa 20 erwartet. Und doch ist die Divergenz dann sehr leicht (?) beweisbar. Derartige Experimente sind besonders wertvoll, da sie zum Widerspruch herausfordern und den darauffolgenden Beweis überhaupt erst relevant erscheinen lassen.

Im Vortrag wurden bewußt nur ganz einfache Beispiele gebracht. APL wurde in dieser Art im Mathematikunterricht in den Schulstufen 3 bis 12 verwendet. Sehr intensiv wird in einer achten Schulstufe damit gearbeitet, wo versucht wird, APL als die bevorzugte mathematische Notation, unterstützt durch den Computer, zu verwenden. Sinnvolle Experimente ergeben sich in vielen Bereichen des Mathematikunterrichtes. Bevorzugte Gebiete besonders in der Oberstufe waren Differential-Integralrechnung, Logik, Gruppentheorie, Matrizenrechnung, Lösung von Gleichungen und Gleichungssystemen.

Literatur:

Berry, Paul et al	APL and Insight, APL Press, 1978
Berry, Paul	SHARP APL Reference Manual, I.P. SHARP, 1979
Iverson, K.E.	Introducing APL to Teachers, APL Press, 1976
Iverson, K.E.	APL in Exposition, APL Press, 1976
Iverson, K.E.	Algebra, An Algorithmic Treatment, Addison-Wesley Publishing Company, 1971
Iverson, K.E.	Elementary Analysis, APL Press, 1976
Iverson, K.E. & Orth, D.L.	CALCULUS in a New Key, APL Press, 1976

Sämtliche Literatur und Information erhältlich über den Autor:

Mag. Gottfried Bach
I.P. SHARP
Rechte Wienzeile 5/3
1040 W i e n
Tel. 57 65 71